Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 6
Quadric Surfaces

Initializations

```
> restart;
  with(plots):
>
```

This section is like an artistic interlude in the semester. Watch and enjoy the fun.

6.1 Plotting of Implicitly Defined Surfaces

We can plot quadric surfaces using the **implicitplot3d** routine in the **plots** package.

Examples

Example 6.1.1

Plot the cone $x^2 + y^2 = z^2$.

Solution

```
> implicitplot3d(x^2+y^2=z^2, x=-3..3, y=-3..3, z=-3..3,
  orientation=[44, 64], axes=boxed);
```



>

Observe that the routine has difficulties about the origin. This can be improved by adding more points using the **numpoints** option. In the next section we will learn how to circumvent the problem by using a parametric representation for this surface.

```
> implicitplot3d(x^2+y^2=z^2, x=-3..3, y=-3..3, z=-3..3,
  orientation=[44, 64], axes=boxed, numpoints=4900);
```



Example 6.1.2

Plot the surface $x^2 + y^2 = z^2 + 1$. This is known as a hyperboloid of one sheet and is the shape of many cooling towers for power plants.

Solution

```
> implicitplot3d(x^2+y^2=z^2+1, x=-4..4, y=-4..4, z=-3..3,
  style=patch, orientation=[44, 64], axes=boxed);
```



>

Example 6.1.3

Plot the surface $x^2+y^2=z^2-2$. This is known as a hyperboloid of two sheets.

Solution

```
> implicitplot3d(x^2+y^2=z^2-2, x=-3..3, y=-3..3, z=-3..3,
  style=patch, orientation=[44, 64], axes=boxed);
```



6.2 Parametric Representation of Quadric Surfaces

A much higher quality of graphics can be obtained if we use parametric representations of the surfaces to be plotted.

Examples

Example 6.2.1

Use elliptic coordinates to plot the hyperboloid of one sheet given by the equation

$$\frac{x^2}{2} + \frac{y^2}{4} = 1 + \frac{z^2}{7}$$

Solution

We use elliptic coordinates that transform the left hand side of this Cartesian equation into a perfect square.

```
> ec:={x=sqrt(2)*r*cos(theta), y=2*r*sin(theta), z=z};
```

$$ec := \{y = 2\, r \sin(\theta),\, z = z,\, x = \sqrt{2}\, r \cos(\theta)\} \qquad (3.1.1.1)$$

Code the Cartesian equation of the hyperboloid and solve for z.

```
> hyp_cart:=x^2/2+y^2/4=1+z^2/7;
```

$$hyp\_cart := \frac{1}{2} x^2 + \frac{1}{4} y^2 = 1 + \frac{1}{7} z^2 \qquad (3.1.1.2)$$

```
> z_cart:=solve(hyp_cart, z);
```

(3.1.1.3)

$$z\_cart := \frac{1}{2}\sqrt{14x^2+7y^2-28}, -\frac{1}{2}\sqrt{14x^2+7y^2-28} \quad (3.1.1.3)$$

Create the x, y, and z coordinates of the top and bottom halves of this hyperboloid.

```
> top:=simplify(subs(ec, [x, y, z_cart[1]]));
```

$$top := \left[\sqrt{2}\, r\cos(\theta), 2\, r\sin(\theta), \sqrt{-7+7r^2}\right] \quad (3.1.1.4)$$

```
> bottom:=simplify(subs(ec, [x, y, z_cart[2]]));
```

$$bottom := \left[\sqrt{2}\, r\cos(\theta), 2\, r\sin(\theta), -\sqrt{-7+7r^2}\right] \quad (3.1.1.5)$$

Finally, we plot both halves in one picture.

```
> plot3d({top, bottom}, r=1..3, theta=0..2*Pi, style=patch,
  orientation=[44, 64], axes=boxed, labels=[x, y, z],
  scaling=constrained, shading=ZHUE);
```




```
>
```

Example 6.2.2

Use cylindrical coordinates to sketch the cone from Example 6.1.1.

Solution

If we substitute $x = r\cos\theta$ and $y = r\sin\theta$ in the equation of the cone $x^2 + y^2 = z^2$, we obtain $z = r$ or $z = -r$. This gives us the following parametric representation for the top half and the bottom half of the cone.

```
> top:=[r*cos(theta), r*sin(theta), r];
```

```
bottom:=[r*cos(theta), r*sin(theta), -r];
```

$$top := [r\cos(\theta), r\sin(\theta), r]$$

$$bottom := [r\cos(\theta), r\sin(\theta), -r] \tag{3.1.2.1}$$

Now, we can create a high quality picture of this cone.

```
> plot3d({top, bottom}, theta=0..2*Pi, r=0..3, axes=boxed,
  orientation=[44, 64], labels=[x,y,z]);
```




```
>
```